

## Listing 2: Arduino Sketch

```
#define led1 13
#define button1 6
#define button2 7
const int NUMBER_OF_FIELDS = 3; // how many comma separated fields we expect
int fieldIndex = 0; // the current field being received
int values[NUMBER_OF_FIELDS]; // array holding values for all the fields

void setup()
{
    // Initialize serial port to same speed as set in e-prime
    Serial.begin(128000);
    pinMode(led1, OUTPUT);
    pinMode(button1, INPUT);
    pinMode(button2, INPUT);
}

void loop()
{
    readTrial();
}

void readTrial()
{
    if( Serial.available())
    {
        char ch = Serial.read();
        if(ch >= '0' && ch <= '9') // is this an ascii digit between 0 and 9?
        {
            // yes, accumulate the value
            values[fieldIndex] = (values[fieldIndex] * 10) + (ch - '0');
        }
        else if (ch == ',') // comma is our separator
        {
            if(fieldIndex < NUMBER_OF_FIELDS-1)
                fieldIndex++; // increment field index
        }
        else
        {
            // any character not a digit or comma ends the acquisition of fields
            runTrial(values[0], values[1], values[2]);
            //clear values again; otherwise they multiply up
            for (int i=0; i <= fieldIndex; i++)
            {
                values[i] = 0;
            }
            fieldIndex = 0; // ready to start over
        }
    }
}

void runTrial(int trialNr, int trialDuration, int trialMessage)
{
    //uncomment following line to see board LED go on during trial
    //digitalWrite(led1, HIGH);

    unsigned long starttime = micros();
    unsigned long endtime = starttime + (unsigned long) trialDuration * 1000;
    unsigned long nowtime;
    int button1State = 0;
    int button2State = 0;

    do
    {
        button1State = digitalRead(button1);
```

```

        button2State = digitalRead(button2);
        nowtime = micros();
    } while (button1State == 0 && button2State == 0 && nowtime < endtime);

    //uncomment if you turned it on above
    //digitalWrite(led1, LOW);    // set the LED off again

    unsigned long rt = nowtime - starttime;
    //constructing answer string
    String buttonStates = "0"; // no response
    if (button1State == 1) {buttonStates = "1";}
    if (button2State == 1) {buttonStates = "2";}
    if (button1State == 1 && button2State == 1) {buttonStates = "3";}

    sendBack(trialNr, rt, buttonStates);
}

void sendBack(int trialNr, unsigned long rt, String message)
{
    Serial.print(trialNr,DEC);
    Serial.print(",");
    Serial.print(rt,DEC);
    Serial.print(",");
    Serial.print(message);
    Serial.print(".");
}

```