

Listing 1: E-Prime InLine script

```
'===== PART 1 - SETUP =====

'variables to save timestamps of start and end
'note: Double can hold values from 4.94066E-324 to 1.797693134862315E308
Dim startTime As Double
Dim postSendingTime As Double
Dim endTime As Double

'variables to be transmitted to Arduino
'note: Integer can hold values from -32768 to 32767
Dim trialNr As Integer
Dim trialDuration As Integer
Dim trialMessage As Integer

'generate random number as check number to verify correspondence of trial
'and what Arduino returns
trialNr = CInt(Random(0,32767))

'reads TrialDuration using the context object.
trialDuration = c.GetAttrib("TrialDuration")

'the following is just reserved, not currently used.
trialMessage = 999

'===== PART 2 - TALKING TO ARDUINO =====

'construct and send String to Arduino.
'Arduino expects 3 Integers separated by a comma and ending
'with a non-integers sign other than a comma
'we End With a dot
Dim sendString As String
sendString = CStr(trialNr) & "," & CStr(trialDuration) & "," &
CStr(trialMessage) & "."

'remember when sent
startTime = Clock.ReadMicrosec

'send
Serial.WriteString sendString

'also save time after sending
postSendingTime = Clock.ReadMicrosec

'print them to debug
'uncomment following for debugging
'Debug.Print sendString

'===== PART 3 - LISTENING TO ARDUINO =====

'now read and wait until the string terminator is in
Dim strData As String
Dim incoming As String
Dim nRead As Long

Dim waiting As Boolean
waiting = True

While waiting
    'read new data and add them to complete string
    nRead = Serial.ReadString(incoming)
    strData = strData & incoming
    'Anything come in yet?
    If Len(strData) > 0 Then
        'last sign of complete string is a . ?
```

```

        If Mid$(strData, Len(strData), 1) = "." Then
            endTime = Clock.ReadMicrosec
            waiting = False
        End If
    End If
Wend

'===== PART 4 - SAVING DATA =====

'print them to debug
'uncomment following line for debugging
'Debug.Print strData

'disassemble string into integers, based on algorithm that is also used
'on the Arduino side
'here we expect 3 fields returning from Arduino: trialNr, RT, buttonStates
Dim NUMBER_OF_FIELDS As Integer
'how many comma separated fields we expect
NUMBER_OF_FIELDS = 3

'the current field being received
Dim fieldIndex As Integer
fieldIndex = 0

'array holding values for all the fields
Dim values(NUMBER_OF_FIELDS) As Long

Dim i As Integer
Dim ch As String
Dim chasc As Integer

For i = 1 To len(strData)
    ch = Mid$(strData, i, 1)
    chasc = Asc(ch)
    'is this an ascii digit between 0 and 9?
    If chasc >= 48 And chasc <= 57 Then
        'yes, accumulate the value
        values(fieldIndex) = (values(fieldIndex) * 10) + ch
        'comma is our separator, so move on to the next field
    ElseIf ch = "," Then
        If fieldIndex < NUMBER_OF_FIELDS-1 Then
            fieldIndex = fieldIndex + 1 'increment field index
        End If
    End If
Next i

'compute a check RT that is the difference between starting the inline
'and receiving the full Return String from arduino - should be longer
'than what RT says
Dim InlineRt As Double
InlineRT = endTime - startTime

'store generated random check number
c.SetAttrib "TrialNr", trialNr

'uncomment following line for debugging
'Debug.Print values(0) & "\t" & values(1) & "\t" & values(2)

'save values returned from Arduino in three attributes of the StimulusList
c.SetAttrib "ArduinoTrialNr", values(0)
c.SetAttrib "ArduinoRT", values(1)
c.SetAttrib "ArduinoButtons", values(2)

'store timings measured by script
c.SetAttrib "InlineStartTime", startTime
c.SetAttrib "InlinePostSendingTime", postSendingTime

```

```

c.SetAttrib "InlineEndTime", endTime
c.SetAttrib "InlineRT", InlineRT

'wait 100 ms to allow the trialStimulus object to finish
'uncomment if you expect reaction times less than the duration
'of the stimulus object
'sleep(100)

'save values also in .RT and .RESP, overwriting the no-response there

'CHANGE NAME OF STIMULUS OBJECT IN FOLLOWING LINES IF YOU
'USE A DIFFERENT ONE
    trialStimulus.RT = values(1)/1000
    trialStimulus.RESP = values(2)

    'now set ACC so that Feedback Display can evaluate it
    'this needs to be done manually here because trialStimulus
    'object is already finished,
    'And would not check the .RESP again to determine ACC
    If trialStimulus.CRESP = trialStimulus.RESP Then trialStimulus.ACC = 1
    If trialStimulus.CRESP <> trialStimulus.RESP Then trialStimulus.ACC = 0

'reset values to zero
For i = 0 To NUMBER_OF_FIELDS-1
    values(i) = 0
Next i
fieldIndex = 0

```